
==========================LORENZ_CHAOS==========================================

MULTI-DIMENSION OSCILLATION WITH RANDOM CROSS-TALK PLOTTED IN 2D.

LORENZ_CHAOS
*                                           C1
*                                     ____||__
*                                    |    ||  |
*                     R1 100K    INN1| |\     |
*                  NY _/\  /\  /\____|_|+\    | X
*                        \/  \/   |    |  \___|_
*                                 |    |  /
*                  X  _/\  /\  /\_|  __|-/ OPA1
*                        \/  \/     _|_|/
*     -XZ/100 = XZ     R2 100K      ///
*          ______                         NY = -Y
*        _|_     |     R3 35.7K
*       /   \    |  X _/\  /\  /\_  INN2    C2
*  X __| \/  |   |       \/  \/   |   ____||__
*      | /\  |   |     R4 10K     |  |    ||  |
*       \___/    |_____/\  /\  /\_|  | |\     |
*  Z _____|              \/  \/   |__|_|+\    | NY
*                      R5 1MEG    |    |  \___|_
*                  NY _/\  /\  /\_|    |  /
*                        \/  \/      __|-/ OPA2
*      -XY/100 = XY                 _|_|/
*          ______                   ///
*        _|_     |     R6 10K               C3
*       /   \    |               INN3 ____||__
*  X __| \/  |   |_____/\  /\  /\_   |    ||  |
*      | /\  |           \/  \/   |  | |\     |
*       \___/                     |__|_|+\    | Z
*  Y _____|        Z  _/\  /\  /\_|    |  \___|_
*                        \/  \/        |  /
*                                    __|-/ OPA3
*                      R7 374K      _|_|/
*                                   ///
*.OPTIONS  GMIN=1p      METHOD=TRAP   ABSTOL=1u     TEMP=27 srcsteps = 1  gminsteps = 1 
*.OPTIONS  RELTOL=.001  ABSTOL=1n     VNTOL=1n      ITL4=500     ITL1=400  
*======== ====== ====== ====== ====== ====== ====== ====== ====== 
VCC       VCC    0      DC     9
VEE       VEE    0      DC     -9

BMULT1    XZ     0      V=    -1*V(X)*V(Z)/10
BMULT2    XY     0      V=    V(X)*V(NY)/10
XOP1      INN1   X      INVOP
XOP2      INN2   NY     INVOP
XOP3      INN3   Z      INVOP
C1        INN1   X      .1u    IC = -1
C2        INN2   NY     .1u    IC = -1
C3        INN3   Z      .1u    IC = -1
R1        NY     INN1   100k
R2        X      INN1   100k
R3        X      INN2   35.7k
R4        XZ     INN2   10k
R5        NY     INN2   1Meg
R6        XY     INN3   10k



R7        Z      INN3   374k

*=========INVERT_OPAMP_MODEL=========================
*
* ___  |\                  |\    ___
*|INN|_| \_/\  /\  /\______| \__|OUT|
*|___| | /   \/  \/  _|_   | /  |___|
*      |/            ___   |/
*                     |
*                    _|_
*                    ///
*
.SUBCKT   INVOP  INN    OUT
 BOPA     OTAOUT 0      V =   9*tanh(tanh(tanh( (-V(INN))*30)*30)*30)
 RLP      OTAOUT LP     500k
 CLP      LP     0      300p
 BOUT     OUT    0      V =   V(LP)
.ENDS     INVOP

.tran      1m    10    0       1m UIC

.control 
run
plot     x  ny  z
plot     ny vs  x
plot     z  vs  x
.endc   
.end

